PIC – mikrokontrollerlariga dastur tuzish asoslari. 
Reja: 
19.1. Laboratoriya maketining bayoni. 
19.2. Mikrokontroller maketini ishga tushirish (initsializasiyalash). 
          19.3. Mashq masalalarini dasturlash. 
 
Tayanch iboralar: laboratoriya maketi, port, tumbler, dasturosti, indikator segmenti, yetti segmentli element, dastur listingi, tovush kengligida o‘zgaruvchi signal, dastlabki holatga o‘tqazish. 
 
19.1. Laboratoriya maketining bayoni. 
Dastlabki mashq dasturlarini tuzish va ularni ishlashini tekshirish uchun, uncha murakkab bo‘lmagan o‘z tarkibiga eng ko‘p tarqalgan tashqi qurilmalarni olgan, maketning bo‘lishi maqsadga mofuq bo‘lar edi. Talabalar tomonidan laboratoriya ishlarini bajarish uchun ishlatiladigan shu kabi maketning sxemasi 19.1-chizmada keltirilgan.  
[image: ]
19.1-chizma. Laboratoriya maketining sxemasi. 
  
V portining RB1 - RB7 yo‘llari umumiy anodli HL1 yetti segmentli indikatoriga xizmat qiladi. Shuning uchun indikator segmentining yorug‘lik tarqatishi 
V portning tegishli chiqishida past qiymatli signal bilan ta’minlanadi. Maket shuningdek sxemada ko‘rsatilmagan dasturlash va kompyuter bilan aloqa vositalariga ega. 
Maket +5V turg‘unlashtirilgan kuchlanish manbasidan ishlaydi. MK takt chastotasi 2 MGs atrofida bo‘lib, RC-zanjir orqali beriladi. A portining RA0 yo‘liga kalit ish tartibiga mo‘ljallab bipolyar tranzistor ulangan, u VA1 tovush karnagiga yuklatilgan. Tovush chiqarish uchun RA0 chiqishiga tovush kengligida o‘zgaruvchi signal berish kerak. A portining RA1 yo‘liga chiqishda yuqori kuchlanish bo‘lganda yorug‘lik tarqatuvchi yorug‘lik diodi VD2 ulangan. SA1 va SA2 tumblerlar (o‘chirib yoqish moslamasi) va shuningdek SB1 va SB2 tugmalar (o‘chirib yoqish moslamasi) mos ravishda A portining RA2 va RA3 yo‘llariga va shuningdek A portining RA4 yo‘liga va V portining RB0 yo‘liga ulangan. Tumbler va tugmalarning boshlang‘ich holati – ulanmagan, bu esa MK ning kerakli kirishlariga yuqori qiymatdagi signal berilishini ta’minlaydi. 
 
19.2. Mikrokontroller maketini ishga tushirish (initsializatsiyalash). 
Foydalanuvchi uchun oddiy dasturlarni yaratishga kirishishdan oldin keyinchalik ishlatiladigan o‘zgaruvchilarni bayon qilish va MK ni ishlashini tanlangan maketga moslash zarur bo‘ladi. Shu maqsadda biz init.asm dastlabki dastur listingini yozamiz va batafsil ko‘pib chiqamiz, uning tarkibiga foydalanuvchining qolgan barcha dasturlari kiritilib boriladi. 
;*************************************************************** ******** 
;*dastlabki dastur listingi 
;***************************************************************
******** 
LIST P=16C84, R=HEX     ;protsessor va sanoq tizim turni aniqlovchi 
direktiva 
               ;sukut saqlash bilan 
;***************************************************************
******* 
;* ishlatiladigan o‘zgaruvchilarni bayon va foydalanuvchining o‘zgaruvchlarini saqlash  
;* uchun yacheyka manzilini tayinlash  
;*************************************************************** ******** 
;  INTCON      EQU        0x0B 
;  OPTION      EQU        0x81 
;  TMR0       EQU        0x01 
;   INTF       EQU        1 
;   T0IF        EQU        5 
   PCL        EQU        0x02 
   STATUS      EQU        0x03 
   RPO        EQU         5 
   PORTA EQU        0x05 
   PORTB EQU        0x06 
   TRISA       EQU        0x05 
   TRISB EQU         0x06 
   W        EQU        0 
   F         EQU        1 
   TEMPA EQU        0x0C 
   TEMPB EQU        0x0D 
   COUNT1      EQU        0x0E 
   COUNT2      EQU        0x0F 
   COUNT3      EQU        0x10 
;*************************************************************** ******** 
;*ma’tinni almashtirish belgilarni aniqlashtirish 
;*************************************************************** ******** 
   #DEFINE  Z     STATUS,2   ;nol natija biti 
   #DEFINE  BA1   PORTA,0   ; BA1 tovush karnagi 
   #DEFINE  VD2   PORTA,1   ;VD2 yorug‘lik diodi 
   #DEFINE  SA1   PORTA,2   ;SA1 tumbler 
   #DEFINE  SA2   PORTA,3   ;SA2 tumbler 
   #DEFINE  SB1    PORTA,4   ;SB1 tugma 
   #DEFINE  SB2    PORTB,0    ;SB2 tugma 
   #DEFINE  HL1_A   PORTB,1    ;A segment- indikatori 
   #DEFINE  HL1_B   PORTB,2    ;B segment- indikatori 
   #DEFINE  HL1_C   PORTB,3    ;C segment- indikatori 
   #DEFINE  HL1_D   PORTB,4    ;D segment- indikatori 
   #DEFINE  HL1_E   PORTB,5    ;E segment- indikatori    #DEFINE  HL1_F   PORTB,6   ;F segment- indikatori 
   #DEFINE  HL1_G   PORTB,7   ;G segment- indikatori 
;*************************************************************** ******** 
;*bajariladigan dastur 
;***************************************************************
******** 
   ORG     0x000     ;nolga o‘tkazish bo‘yicha boshlang‘ich manzilni o‘rnatish 
   GOTO    BEGIN    ;dastur boshlanishiga o‘tish 
   ORG   0x005     ;dastur joylashtirishda boshlang‘ich manzilni o‘rnatish  
   BEGIN 
  CALL INIT_PORTS   ;MK portlarini nol holatga o‘tkazish dasturostini chaqirish 
;*************************************************************** *********** 
;*foydalanuvchining dasturi 
;*************************************************************** *********** 
; 
INIT_PORTS         ; portlarini nol holatga o‘tkazish dasturosti 
  MOVLW   0xFF     ; A va B portlar yo‘lini bir holatiga 
  MOVWF   PORTA    ; o‘rnatish 
  MOVWF   PORTB    ; 
  BSF     STATUS,RPO ;1 bankka o‘tish 
  MOVLW   0x1C     ; A portining RA0 va RA1 chiqarishga- qolganlari  
  MOVWF   TRISA    ; kiritishga 
  MOVLW   0x01     ;RB0 yo‘lini sozlash 
  MOVWF   TRISV    ;V porti kiritishga-qolganlari chiqarishga 
  BCF     STATUS,RPO ;0 bankka qaytish 
  RETURN          ;dasturostidan qaytish 
 ; 
  END           ;dastur tugadi 
  
Bu dasturning ishlashini ko‘rib chiqamiz. Boshida u assemblerga ishlatiladigan MK turini va sukut saqlash bilan sanoq tizimini ham ko‘rsatadi. Undan so‘ng keluvchi EQU assembler derektivalari bu va keyingi dasturlarda ishlatiladigan assembler konstantalarini aniqlaydi. Ular aniq MK tarkibiga bog‘langan aniq lekin murakkab assembler ifodalarining o‘rniga dastur matnida ancha qulay mnemonik belgilarni ishlatish imkonini beradilar. TEMPA, TEMPB, COUNT1 va COUNT2 ko‘rsatuvchilari oraliq axborotlarni saqlash uchun xotira yacheykasining manzilini belgilaydi (hozirdagi holat, o‘zgaruvchan sikllar va hokazo). 
   #DEFINE assembler direktivalari mos belgini o‘rnini bosadigan qatorni har gal u dastlabki matnda uchratsagina beradi. Bizning holimizda bu direktivalar maket sxemasiga bog‘langan belgili nomlarni port va registrlarning tegishli razryadlarining jismoniy manzili o‘rnida ishlatishga imkon beradi. Bunda shuni nazarda tutish kerakki, #DEFINE direktivasi bilan aniqlangan belgilar simulyator orqali ko‘rib bo‘lmaydi. Shuning uchun ko‘rishga port va registrlarning jismoniy manzilini ishlatish zarurdir. 
ORG 0x00 direktivasi dasturiy kodni boshlanish manzilini nolga teng qilib o‘rnatadi, yaʻni MK ning buyruqlar sanoq qurilmasini nolga o‘tkazilgandan keyingi holatiga mos holatga. GOTO BEGIN buyruqlari ORG 0x005 assembler direktivasi va BEGIN belgisi bilan birgalikda dastur xotirasining 0x005 manziliga o‘tishni ta’minlaydi, shu manzildan boshlab dasturning asosiy qismi joylashadi. Bu uzilish vektori sifatida ishlatiladigan 0x004 manzilini aylanib o‘tish uchun zarur, shunday qilib uni keyingi ishlatilishi mumkin bo‘lgan hollarga zahiralanadi. 
So‘ng CALL INIT_PORTS buyrug‘i yordamida portlarni dastlabki holatga o‘tkazish dasturostini chaqirish amalga oshiriladi. Dastlabki holatga o‘tkazish dasturi birinchi bo‘lib axborotlarni chiqish triggeri holatini yuqori (birlik ) holatga o‘rnatadi. Bu operatsiya mikrokontroller yaratuvchilariga port registrlarining noaniq holatini yo‘q qilish uchun taklif etiladi. So‘ng BSF STATUS,RP0 buyrug‘i orqali axborotlar xotirasining 1 bankiga ulanishni amalga oshiriladi, u yerda TRISA va TRISB axborotlarni uzatilish yo‘nalishini boshqarish registrlari joylashgandir. MOVLW 0x1C va MOVWF TRISA buyruqlari yordamida A portining RA0 va RA1 yo‘llari chiqarishga, qolganlari esa kiritishga sozlanadi. MOVLW 0x01 va MOVWF TRISTB buyruqlari yordamida B portining RB0 yo‘lini kiritishga, qolganlarini esa chiqarishga sozlanadi. BCF STATUS,RP0 buyrug‘i yordamida 0 bankka qaytish amalga oshiriladi, u yerda dastur ishlashi uchun zarur bo‘lgan portlar va registrlar joylashgan bo‘ladi. 
Maket bilan ishlash jarayonida portlarni qayta sozlash amalga oshirilmaganligi uchun va kiritilgan o‘zgaruvchilar barcha mashq masalalari uchun yetarli bo‘lganligi tufayli ular keyinchalik sukut saqlash bo‘yicha dastlabki dastur init.asm ta’rkibiga kiritilgan deb qaraladi. Mashq masalalarini yozishda imkon boricha tarkibiy dasturlash usulini ishlatishga xarakat qilamiz, unda amaliy dastur qandaydir dastur modullar to‘plami yordamida quriladi, ularning har biri axborotga ishlov berishning ma’lum protseduralarini amalga oshiradi. Shu bilan bir qatorda har bir dasturiy modul faqat bitta kirish nuqtasiga va bitta chiqish nuqtasiga ega bo‘ladi. Bir marotaba qachondir kiritilgan dasturiy modul o‘z nomi bilan boshqa amaliy dasturlarda ham ishlatilishi mumkin. 
 
19.3. Mashq masalalarini dasturlash. 
Mashq masalalarini dasturlashni SB1 tugmasining holatini o‘qiydigan va uni VD2 yorug‘lik diodiga chiqarish dasturni yozishdan boshlaymiz. Tugmaning bosilmagan holati (RA4 kirishdagi signalning yuqori qiymati) yorug‘lik diodining yorug‘lik tarqatish xolatiga to‘g‘ri keladi va aksi. 
  ;asosiy dastur 
  LOOP 
    CLRWDT           ; 
 CALL    GET_RA     ; GET_RA  
    CALL    SB1_VD2     ; SB1_VD2   
    GOTO    LOOP       ; LOOP  
  ; 
  GET_RA              ; 
    MOVF    PORTA,W     ;A W 
    MOVWF   TEMPA      ;W TEMPA 
    RETURN            ; 
  ; 
  SB1_VD2              ; 
    BTFSS TEMPA,4        ; TEMPA,4=1 
    GOTO    P0        ;P0 
    BSF     VD2       ;VD2 
 P0 
    BTFSC TEMPA,4        ; TEMPA,4=0 
    GOTO    P1        ; P1 
    BCF     VD2       ; 
 P1  
     RETURN 
 ; 
Asosiy dastur LOOP – GOTO LOOP yopiq siklni o‘z ichiga oladi, bu esa tugma holatini davriy ravishda nazorat qilish va uning holatini yorug‘lik diodiga chiqarib turishni siklik ravishda qaytalab turish uchun kerak bo‘ladi. CLRWDT buyrug‘i dastur ishini qo‘riqchi taymerning to‘lishi bo‘yicha tasodifiy nolga o‘tish holatini bartaraf qiladi. Ikkita keyingi buyruqlari GET_RA va SB1_VD2 dasturostilarni chaqirishni amalga oshiradi. Ulardan birinchisi (GET_RA) boshida A portining hozirdagi holatini o‘qiydi, o‘qilgan qiymatlarni ishchi W registriga joylashtiriladi. Ishchi registr boshqa buyruqlarni bajarilishida kerak bo‘lib qolishi mumkin bo‘lganligi uchun uning qiymati bu yerda A portining holatini vaqtinchalik saqlash uchun ishlatiladigan TEMPA registriga yoziladi. Shunday qilib, GET_RA dasturostidan qaytgach TEMPA registrining 4 razryadida SB1 tugma holati haqidagi axborot bo‘ladi: “1” bo‘lganda tugma bosilmagan, “0” bo‘lganda esa tugma bosilgan bo‘ladi.        
SB1_VD2 dasturosti TEMPA registrining 4 razryadi holatini tahlil qilib va uning natijasiga qarab yorug‘lik diodlarini yoqadi yoki o‘chiradi. PIC16F84 MK buyruqlar tizimida shartli o‘tish buyruqlari yo‘q, shuning uchun u yoki bu shartni tekshirishni tashkil etish uchun berilgan registrning (BTFSS va BTFSC) ma’lum bitlarining holatiga qarab dasturning keyingi buyrug‘ini bajarilishini o‘tkazib yubora oldadigan buyruq ishlatiladi. Xususan, BTFSS TEMP,4 buyrug‘i GOTO P0 buyrug‘ini bajarilishini o‘tkazib yuboradi, agarda TEMP,4=1 bo‘lsa (tugma bosilmagan bo‘lsa). Shu bilan bir qatorda BSF VD2 buyrug‘i joriy etiladi, u VD2 yorug‘lik diodini yoqadi. So‘ng TEMP,4=0 (tugma bosilgan) shart tahlil qilinadi va agarda uning o‘rni bo‘lsa yorug‘lik diodi o‘chiriladi.  
Berilgan algoritmning ancha sodda yechimi ham bo‘lishi mumkin, chunki tugmaning bosilgan holat bosilmagan holatini inkor qiladi (va aksi), lekin keltirilgan variant ancha tushunishga oson.  
Dasturning ozroq murakkablashgan variantini ko‘rib chiqamiz, VD2 yorug‘lik diodini yoqilishini faqat tumbler va tugmalarning quyidagi holatlaridagina soir bshladi deb nazarda tutiladi: SA1=1, SA2=1, SB1=1 va SB2=0.  
 ;asosiy dastur 
   LOOP 
   CLRWDT        ;qo‘riqchi taymerni nolga o‘tkazish 
   CALL    GET_RA   ;GET_RA dasturostini chaqirish  
   CALL    GET_RB   ;GET_RB dasturostini chaqirish 
   CALL    ZAG_1110  ;ZAG_1110 dasturostini chaqirish    
   GOTO   LOOP    ;jarayonni qaytarish uchun LOOP belgisiga o‘tish 
 ; 
 GET_RB           ;V port holatini o‘qish dasturostisi  
   MOVF   PORTB,W   ;W dan B port holatini o‘qish 
   MOVWF  TEMPB    ;W ni TEMPB ga uzatish 
RETURN 
 ; 
 ZAG_1110        ;VD2 yorug‘lik diodini faqat maket tumbler va   
                ;tugmalarining quyidagi holatida yoqadi 
              ;SA1=SA2=SB1=1 va SB=0 
    BTFSS TEMPA,2    ;agarda TEMPA,2=1 buyruqni o‘tkazib yubor        
    GOTO     P0    ; 
    BTFSS TEMPA,3      ;agarda TEMPA,3=1 buyruqni o‘tkazib yubor                  GOTO     P0     ; 
    BTFSS TEMPA,4      ; agarda TEMPA,4=1 buyruqni o‘tkazib yubor                  GOTO     P0     ; 
    BTFSC TEMPB,0      ; agarda TEMPV,0=0 buyruqni o‘tkazib yubor                  GOTO     P0     ; 
    BSF      VD2    ; VD2 yorug‘lik diodi yoqilsin 
    GOTO     P1 
  P0 
    BCF      VD2    ;VD2 yorug‘lik diodi o‘chirilsin 
  P1 
    RETURN 
  ; 
    INCLUDE   GET_RA.ASM 
  ;  
GET_RA va GET_RB dasturostilari TEMPA va TEMPB registrlariga A va V portlarning hozirdagi holatini joylashtiradilar. ZAG_1110 dasturosti TEMPA registrning 2,3 va 4 razryadlar holatini va TEMPB registrining 0 razryadini tahlil qiladi. So‘ng TEMPA,2,3,4=1,1,1 va TEMPB,0=0 sharti bajarilsa VD2 yorug‘lik diodini yoqadi. Yuqoridagi shartlardan birortasi bajarilmasa yorug‘lik diodi o‘chadi. 
INCLUDE GET_PORTA.ASM diriktivasini ishlatilishi sozlanib bo‘lgan dasturosti modullarini hozirda ishlatilayotgan dasturga kiritish imkonini beradi. Bu imkoniyatdan foydalanish mumkin bo‘lishi uchun, sozlangan modullarni alohida assembler fayllari ko‘rinishida saqlash zarur. 
Endi yetti segmentli indikatorni maket tumblerlarining holatini nazorat qilishi uchun ishlatishga harakat qilib ko‘ramiz. Dastlab dastur yozamiz, u HL indikatoriga har qanday ikkilik 0b dan 1111b gacha sonining o‘n oltilik sanoq tizimidagi ifodalanishini yetti segmentli tasvirga chiqaradigan bo‘lishi kerak. 
  ;asosiy dastur 
   LOOP 
   CLRWDT        ;qo‘riqchi taymerni nolga o‘tkazish 
   MOVLW0x0A       ;0A konstantani W ga uzatish 
   CALL    SEV_SEG  ; SEV_SEG dasturostini chaqirish    
   MOVWF   PORTB   ; PORTB ga Wni jo‘natish 
   GOTO   LOOP    ;jarayonni takrorlash uchun LOOP belgisiga o‘tish     
 ; 
 SEV_SEG          ;yetti segmentli indikatorga xizmat  
               ; ko‘rsatish dasturostisi 
   ANDLW0x0F       ; W ni 4 ta kichik razryadini maskalash va 4 ta  
               ; katta razryadlarini nolga o‘tkazish 
   ANDWF PIC,F      ; W ni PCL bilan qo‘shish va natijani PCL ga  
               ; jo‘natish 
   RETLW0x80       ;dasturostini 80 dan W ga qaytarish 
   RETLW0xF2       ;dasturostini F2 dan W ga qaytarish  
   RETLW0x48       ;dasturostini 48 dan W ga qaytarish  
   RETLW0x60       ;dasturostini 60 dan W ga qaytarish 
   RETLW0x32       ;dasturostini 32 dan W ga qaytarish 
   RETLW0x25       ;dasturostini 25 dan W ga qaytarish 
   RETLW0x04       ;dasturostini 04 dan W ga qaytarish 
   RETLW0xF0       ;dasturostini F0 dan W ga qaytarish 
   RETLW0x00       ;dasturostini 00 dan W ga qaytarish 
   RETLW0x20       ;dasturostini 20 dan W ga qaytarish 
RETLW0x10       ;dasturostini 10 dan W ga qaytarish 
   RETLW0x06       ;dasturostini 06 dan W ga qaytarish 
   RETLW0x8C       ;dasturostini 8C dan W ga qaytarish 
   RETLW0x42       ;dasturostini 42 dan W ga qaytarish 
   RETLW0x0C       ;dasturostini 0C dan W ga qaytarish 
   RETLW0x1C       ;dasturostini 1C dan W ga qaytarish 
; 
Dastur o‘z ishini 0x0A konstantani W ishchi registrga jo‘natishdan boshlaydi. So‘ng yetti segmentli indikatorga xizmat ko‘rsatuvchi SEV_SEG dasturostisini chaqirish amalga oshiriladi. SEV_SEG dasturostisini ishlashi W ishchi registrining 4 ta kichik razryadini maskalashdan va 4 ta katta razryadlarini nolga o‘tkazishdan boshlanadi. Shunday qilib W ishchi registridan uzatilayotgan sonning katta razryadlari tahlilda ishtirok etmaydi. So‘ng W registrining maskalashtirilgan qiymatlarini PCL buyruqlar sanoq qurilmasining kichik baytining hozirdagi holatiga qo‘shiladi va natija PCL ga joylashtiriladi. Shunday qilib, buyruqlar sanoq qurilmasida qo‘shimcha surilish hosil qilinmoqda, uning kattaligi ishchi registrga uzatilgan qiymat bilan teng. Masalan, agarda W=0 ga teng bo‘lgan bo‘lsa, u holda buyruqlar sanoq qurilmasining qiymati o‘zgarmaydi va keyingi RETLW 0x80 buyrug‘ini bajaradi, u 0x80=B’1000000’ yozuvli dasturostidan W registriga qaytishni chaqiradi. Agarda, keltirilgan dasturda bo‘lganidek, W=0A bo‘lsa u holda PCL qiymatiga 0x0A soni qo‘shiladi va 10 qadamga qo‘shimcha surilish sodir bo‘ladi. 
Natijada RETLW 0x10 buyrug‘i bajarilgan bo‘ladi, u 0x10=B’0001000’ yozuvli dasturostidan W registriga qaytishni chaqiradi. 
Dasturostidan qaytilgandan so‘ng W ni PORTB ga jo‘natish sodir bo‘ladi va uning holati yetti segmentli HL indikatorida aks ettiriladi. Xususiy holda, agar W=0 bo‘lsa, unda 1000000b ni V portiga chiqarilsa yetti segmentli indikatori 0 ko‘rsatadi, W=A bo‘lganda esa O ko‘rsatadi. Shunday qilib, har qanday 4 razryadli ikkilik sonini indikatorda aks ettirish mumkin. 
SEV_SEG dasturostisida ishlatilgan buyruqlar sanoq qurilmasini toʻg‘ri boshqarish usulini sonlarni jadvalli konvertatsiyalashni joriy etish uchun ishlatlishi mumkin. Bu holda shuni eʻtiborda tutish kerakki , ushbu usul bitta jadvalda 256 ta qiymatdan ortig‘ini konvertatsiyalashga imkoni yo‘q. Undan tashqari buyruqlar sanoq qurilmasining kichik baytini to‘lishini oldini olish uchun jadvalli konvertatsiyalash dasturi 256- baytli blok ichiga butunligicha sig‘ishi kerak bo‘ladi. 
Endi SEV_SEG dasturostini ishlatib dastur tuzamiz, u SA1 va SA2 tumblerlar holatlarini o‘qib va uni indikatorga mos son tariqasida chiqaradi. 
  ;asosiy dastur 
   LOOP 
   CLRWDT        ;qo‘riqchi taymerni nolga o‘tkazish 
   CALL   GET_RA   ;GET_RA dasturostini chaqirish     
   RRF    TEMPA,F  ;o‘tish orqali bir razryad o‘nga surish 
   RRF    TEMPA,W  ; o‘tish orqali bir razryad o‘nga surish 
   ANDLW0x03      ;ikkita kichik razryadga maska 
    CALL   SEV_SEG  ;SEV_SEG dasturostini chaqirish  
    MOVWF  PORTB  ;W ni PORTB ga jo‘natish 
    GOTO   LOOP   ;jarayonni takrorlash uchun LOOP belgisiga o‘tish 
 ; 
    INCLUDE  GET_RA.ASM 
    INCLUDE  SEV_SEG.ASM  
 ;  
GET_RA dasturostisi TEMPA registriga A portining hozirdagi holatini joylashtiradi. Shunday qilib, TEMPA registrining 2 va 3 – razryadlarida SA1va SA2 tumblerlarning hozirdagi holatlari saqlanadi. Tumblerlarning holat bitlari 0 va 1 holatlarni TEMPA registrida egallashi uchun o‘tish orqali o‘nga ikkita surish amalga oshiriladi, ikkinchi surishning natijasi W registriga joylashtiriladi. So‘ng ishchi registrning ikkita kichik razryadga maska qo‘yiladi va SEV_SEG dasturostini chaqirish amalga oshiriladi. Dasturostidan chiqilgandan keyin natija V portga beriladi va indikatorda aks ettiriladi. 
Endi real vaqt o‘lchamida ishlaydigan, yaʻni ma’lum davomiylikdagi va ma’lum chastotali signallarni xosil qiluvchi yoki kirish signallarining vaqt ko‘rsatgichlarini hisobga oluvchi dasturlarni ko‘rib chiqamiz. Bunday dasturlarning asosiy elementi bo‘lib vaqt ushlanishlarini hosil qiluvchi dasturosti xizmat qiladi. Bo‘lishi mumkin bo‘lgan bunday dasturosti variantlaridan birini ushlanishlarni hosil qilishni dasturiy usullarini qo‘llash orqali ko‘rib chiqamiz, yaʻni joylashtirilgan taymerni ishlatmasdan. 
  ;asosiy dastur 
    MOVLW0xL        ;H’L’ konstantalarni W jo‘natish 
    CALL    DELAY    ;DELAY dasturostini chaqirish 
  ;  
    DELAY          ;vaqt ushlanishini hosil qilish dasturosti 
    MOVWF   COUNT1    ;COUNT1 registrga W ni yuklash 
   LOOPD 
    DECFSZ   COUNT1, F  ;COUNT1 dekrementi 
    GOTO    LOOP     ;siklni H’L’ marotaba takrorlash 
    RETURN          ;dasturostidan qaytish 
  ;  
Asosiy dastur DELAY dasturostini W ishchi registrida qandaydir L konstanta bilan chaqirishni amalga oshiradi, bu dasturostini ichki sikllar sonini aniqlaydi. DELAY dasturosti o‘z ishini ishchi registr qiymatini foydalanuvchining COUNT1 registriga yuklashdan boshlaydi. DECFSZ COUNT1, F buyrug‘i COUNT1 registr qiymatini birga kamaytiradi va uni nolga tengligiga tekshiradi. COUNT1 registrining nolga tengligi sikldan chiqishga va dasturostidan qaytishga olib keladi. Har bir ichki siklni bajarish uchun MK ning uchta mashina sikli zarur bo‘ladi (1-sikl DECFSZ buyrug‘ini nol natijada bajarishga va 2 ta sikl har bir GOTO buyrug‘ini bajarish uchun). DELAY dasturostidan chiqish 4 ta siklni talab etadi (2 ta sikl DECFSZ buyrug‘ini nol natijada bajarishga va 2 ta sikl RETURN uchun). Agar bunga ishchi registrga konstantalarni yuklash, dasturostini chaqirish va COUNT1 foydalanuvchining registriga yuklash uchun zarur bo‘lgan 4 ta sikl qo‘shilsa, u holda DELAY dasturostining bajarilishini umumiy (ushlanish) vaqti tashkil etadi 
      𝑇𝐷 = 4+ 3*(L - 1) + 4 = 5 + 3*L sikllar, 
bu yerda L – konstanta, DELAY dasturostisiga ishchi registr orqali uzatilgan. 
  𝑓𝑜𝑠𝑐= 2 MGs takt chastotasida sikl vaqti 𝑡𝑠= 2 mks ga teng, shuning uchun L= H’00’ =.0 yuklashda maksimal vaqt oralig‘i (interval) 1,55 ms ni tashkil etadi. 
Bunday natija DECFSZ buyrug‘i avval registr qiymatini (H’00’ – 1 = H’FF’) dekrementlashi va undan so‘nggina natijani tahlillash bilan bog‘liqdir.  
Minimal vaqt oralig‘ini ham shu shartlar, 5 sikllar yoki 10 mks bilan hosil qilinadi. Bunday oraliqni olish uchun DELAY dasturostini chaqirishdan oldin ishchi registriga 0x01 sonini yuklash zarur. 
Hosil qilinadigan vaqt oraliqlarining yuqori chegarasini kengaytirish va shuningdek dasturosti bilan ishlashni osonlashtirish uchun LOOPD sikliga bitta yoki bir nechta buyruq qo‘shish mumkin, uning uchun ko‘pincha NOP buyrug‘i ishlatiladi. Misol tariqasida vaqt ushlanishini hosil qiluvchi DELAY_C dasturostini ko‘rib chiqamiz. 
 ; 
     DELAY_C       ;vaqt ushlanishini hosil qilish dasturosti  
                ;(S varianti) 
     MOVWF   COUNT1  ; COUNT1 registrga W ni yuklash 
  LOOPD 
     NOP          ;bo‘sh buyruq 
     DECFSZ   COUNT1, F ; COUNT1 dekrementi   
     GOTO    LOOPD   ;siklni H’L’marotaba takrorlash 
     RETURN        ;dasturostidan qaytish 
  ; 
DELAY_C dasturostini ishlatilishining umumiy vaqti uni chaqirish vaqtini qo‘shib hisoblaganda 
         𝑇𝐷 = 4 + 4*(L - 1) + 4 = 4 + 4*L siklni tashkil etadi. 
   𝑓𝑜𝑠𝑐= 2 MGs takt chastotasida va L = H’F9’ = .249 konstantani yuklashda hosil qilinadigan vaqt oralig‘i rosa 2 ms ni tashkil etadi. Konstantani birga kamaytirish bilan hosil qilinadigan vaqt oralig‘ini 8 mks ga kamaytiriladi. Xususiy holda, L = .124 bo‘lganda 1 ms li ushlanish hosil bo‘ladi. 
Katta vaqt oraliqlarini hosil qilish uchun (yaʻni sekundning ulishlari orasida yotuvchi) bunday yondoshish noqulay. U holda joylashtirilgan sikllar ishlatiladi, quyida berilgan dasturda keltirilganidek. 
  ;asosiy dastur 
    MOVLW0xL        ; H’L’ konstantani W ga jo‘natish 
    CALL    DELAY_D  ; DELAY_D dasturostini chaqirish   
  ; 
  DELAY_D          ;katta vaqt ushlanishlarini hosil qilish 
                 ;dasturosti 
   (D varianti) 
     MOVWF   COUNT2   ;COUNT2 registriga W ni yuklash 
     CLRF    COUNT1  ;COUNT1 registr qiymatini nolga o‘tqazish 
   LOOPD 
     DECFSZ   COUNT1, F  ;COUNT1 ni dekrementi 
     GOTO    LOOPD   ;siklni 256 marotaba takrorlash 
     CLRWDT         ;qo‘riqchi taymerni nolga o‘tkazish 
     DECFSZ   COUNT2,F  ; COUNT2 ni dekrementi 
     GOTO    LOOPD   ;siklni H’L’ marotaba takrorlash 
     RETURN         ;dasturostidan qaytish 
   ; 
DELAY_D dasturostining ichki siklni bajarilish vaqti 3*256 + 4 MKning mashina siklidan tashkil topgan, shuning uchun umumiy ushlanish           𝑇𝐷 = 5 + (3*256 + 4)*L sikldan tashkil topadi. 
  𝑓𝑜𝑠𝑐= 2 MGs takt chastotasida sikl vaqti 𝑡𝑠= 2 mks ga teng, shuning uchun L = 
H’00’ = .0 ni yuklashda maksimal hosil qilinadigan vaqt oralig‘i 0,4 s atrofida bo‘ladi. 
Hosil qilingan vaqt oralig‘i yetarli darajada katta bo‘lganligi uchun tashqi siklga qo‘riqchi taymerni nolga o‘tkazish buyrug‘i kiritilgan. 
0,4 s vaqt oralig‘i sekundga nisbatan marta vaqt oraliqlarni olishga noqulay, shuning uchun yana bitta katta vaqt oraliqlarni hosil qilish dasturini ichki sikliga 
NOP qo‘shimcha buyrug‘ini kiritilgan variantini ko‘rib chiqamiz. 
  ; 
  DELAY_E           ;katta vaqt ushlanishini hosil qilish  
                 ; dasturosti 
  (E varianti) 
    MOVWF   COUNT2   ;COUNT2 registrga W ni yuklash 
    CLRE    COUNT1   ;registr qiymatini nolga o‘tkazish     
    LOOPD 
    NOP           ;bo‘sh buyruq 
    DECFSZ   COUNT1, F  ;COUNT1 dekrementi 
    GOTO    LOOPD    ;siklni 256 marotaba takrorlash 
    CLRWDT         ;qo‘riqchi taymerni nolga o‘tkazish 
    DECFSZ   COUNT2, F  ;COUNT2 dekrementi 
    GOTO    LOOPD    ;siklni H’L’marotaba takrorlash 
    RETURN         ;dasturostidan qaytish 
  ; 
DELAY_E dasturostining ichki siklini bajarilish vaqti 4*256 + 4 MK mashina siklini tashkil etadi, shuning uchun umumiy ushlanish 
           𝑇𝐷 = 5 + (4*256 + 4)*L sikldan tashkil topadi. 
𝑓𝑜𝑠𝑐= 2 MGs takt chastotasida va L = H’F3’ = .243 ni yuklashda hosil qilinadigan vaqt oralig‘i 0,5 s atrofida bo‘ladi, xatolik esa 0,2% ko‘p emas. Agarda ancha yuqori aniqlik zarur bo‘lsa, ushlanish hosil qilinadigan tashqi siklga keraklicha bo‘sh buyruqlarni joylashtirish mumkin bo‘ladi. 
Quyida vaqt ushlanishini hosil qiluvchi dasturostini ishlatib bir necha dasturlarni ko‘rib chiqamiz. VA1 tovush chiqaruvchi karnakni SB1 tugmasi bosilganda tovush signalini beruvchi dasturni yozishdan boshlaymiz. Tovush karnagi faqat RA0 chiqishiga davriy o‘zgaruvchi signal berilgandagina tovush chiqaradi. Tovush yaxshi eshitilishi uchun uning chastotasi maksimal ravishda inson qulog‘i eshitadigan chastota atrofida bo‘lishi kenak. Tovush tarqalish chastotasini 1 KGs ga teng qilib tanlaymiz, bu esa 1 ms impulslarni kelish davriga mos tushadi. 
;asosiy dastur 
  LOOP                   
    CLRWDT       ;qo‘riqchi taymerni nolga o‘tkazish 
    CALL   GET_RA  ;dasturostini chaqirish 
  GET_PORTA 
    CALL   SB1_BA1  ;SB1_BA1 dasturostini chaqirish 
    GOTO   LOOP   ;jarayonni takrorlash uchun LOOP belgiga o‘tish 
  ; 
  SB1_BA1          ;SB1 tugma bosilganda BA1 tovush karnagiga  
               ;tovushni berish  
    BTFSC   TEMPA,4  ;agarda TEMPA,4=0 (tugma bosiq) bo‘lsa  
               ;buyruqni o‘tkazib yuborish 
    GOTO    B0    ;B0 ga o‘tish 
    BSF     BA1    ;RA0 ga yuqori qiymatni berish 
    MOVLW0x3E      ;H’3E’ = .62 konstantalarni Wga jo‘natish 
    CALL    DELAY_C  ;DELAY_C dasturostini chaqirish  
    BCF     BA1    ;RA0 ga past qiymatni berish 
    MOVLW0x3E      ;H’3E’ = .62 konstantalarni Wga jo‘natish  
    CALL    DELAY_C  ; DELAY_C dasturostini chaqirish 
  B0 
    RETURN 
  ; 
    INCLUDE  GET_RA.ASM 
    INCLUDE  DELAY_C.ASM 
  ; 
Avvalgi dasturdagi kabi, GET_RA dasturosti A portning hozirdagi holatini o‘qiydi, so‘ng uni TEMPA registriga uzatadi. SB1_BA1 dasturosti TEMPA registrining 4 razryadini holatini tahlil qiladi va tahlil natijasiga qarab BA1 tovush karnagini tovush signali bilan ta’minlaydi yoki ta’minlamaydi. RA0 yo‘lini birlik va nollik holatda keraklicha ushlab turishni DELAY_C dasturosti L = H’2E’ =.62 ko‘rsatgich bilan ta’minlaydi. Bu esa 0,5 ms atrofidagi ushlanish vaqtiga mos keladi, qilingan ish natijasida kerakli 1 KGs signalni keliq chastotasi hosil qilinadi. 
Quyida SB1 tugmasiga bosilganda VD2 yorug‘lik diodini o‘chib yonishga majbur etuvchi dasturni ko‘rib chiqiladi. O‘chib yonish yaxshi ko‘rinishi uchun uning chastotasini 1Gs ga teng qilib olamiz. 
  ;asosiy dastur 
  LOOP 
    CLRWDT        ;qo‘riqchi taymerni nol holatga o‘tkazish 
     CALL  GET_RA   ;GET_RA dasturostini chaqirish  
     CALL  SB1_VD2M  ;SB1_VD2M dasturostini chaqirish  
     GOTO  LOOP    ;jarayonni takrorlash uchun LOOP belgiga o‘tish 
  ; 
  SB1_VD2M         ;SB1tugmaga bosilganda VD2 yorug‘lik  
               ;diodini o‘chib yonishini ta’minlash dasturosti 
     BTFSC  TEMPA,4  ;agardaTEMPA,4=0 (tugma bosiq) bo‘lsa  
                ;buyruqni o‘tkazib yuborish 
     GOTO  V0     ;V0 ga o‘tish 
     BSF   VD2    ;VD2 yorug‘lik diodini yoqish 
     MOVLW0xF3      ;H’F3’ = .243 konstantalarni Wga jo‘natish 
     CALL   DELAY_E   ;DELAY_E dasturostini chaqirish  
     BCF   VD2     ;yorug‘lik diodini o‘chirish 
     MOVLW0xF3      ;H’F3’ = .243 konstantalarni Wga jo‘natish  
     CALL   DELAY_E   ;DELAY_E dasturostini chaqirish  
   V0 
     BTFSS TEMPA,4    ;agarda TEMPA,4 =1 (tugma bosilmagan) bo‘lsa  
                ;buyruqni o‘tkazib yuborish 
     GOTO   V1     ;V1 ga o‘tish 
     BCF   VD2     ;yorug‘lik diodini o‘chirish 
   V1 
     RETURN 
   ;  
     INCLUDE  GET_RA.ASM 
     INCLUDE  DELAY_E.ASM 
   ;    
Dastur deyarli oldingi ko‘rib chiqilgan dastur bilan bir xil ishlaydi. Birinchi farqli tomoni shundan iboratki, yorug‘lik diodlari majburiy ravishda tugmaning bosilmagan holatida ham o‘chiriladi. Ikkinchi farqi shundan iboratki bu yerda vaqt oralig‘ining kattaligi 0,5 s ni tashkil etadi va uni DELAY_E dasturosti hosil qiladi. 
Vaqt ushlanishlarini hosil qiluvchi dasturostilar quyidagi tashqi signal manbalari bilan ishlaganda foydali bo‘lishi mumkun, tumbler, tugma, o‘chirib yoquvchi moslamalar va hakazo. Gap shundaki , barcha mexanik ulovchi moslamalar bitta yomon ta’sir etuvchi kamchilikka egalar, kontaktlarni “titrash”i deb nom olgan, uning hosil bo‘lishiga kontaktlarning ulanib uzilishidagi mexanik tebranishlar sababchidir. Tebranishlar davri odatda bir necha milli sekundni tashkil etadi, bu vaqt oralig‘ida MK kirishiga ideal impuls o‘rniga impuls to‘plamlari kelishi mumkin.  
Apparat usulda “titrashi” ga qarshi kurashish RS-triggerlarini, bir tebratuvchilarni yoki Shmitt triggerini ishlatish orqali olib boriladi. MK asosidagi qurilmalarda kontakt “titrash” ini bosish uchun odatda dasturiy usul bilan amalga oshiriladi, u port yo‘lini holatini ma’lum vaqtdan so‘ng qayta o‘qishga asoslangan. 
Misol tariqasida A port holatini “titrashsiz” o‘quvchi dasturosti variantini ko‘rib chiqamiz. 
  ; 
  GET_RAD           ;A port holatini o‘qishni “titrashni” bosish                   ;dasturostisi TEMPA registriga  
  DD 
    MOVF    PORTA,W  ;A port holatini W da o‘qish 
    ANDLW0x1C       ;W ishlatilmaydigan bitlariga b’00011100’  
                ;maskasini qo‘yish  
    MOVWF   TEMA    ;W ni TEMA ga jo‘natish 
    CLRWDT         ;qo‘riqchi taymerni nolga o‘tkazish 
  WDT 
    MOVLW0x0A       ;H’0A’ = .10 konstantani W ga jo‘natish 
    CALL     DELAY_E  ;DELAY_E dasturostini chaqirish     
    MOVF    PORTA,W  ;A port holatini W da o‘qish  
    ANDLW0x1C        ;W ga b’00011100’maskani qo‘yish 
    SUBWF TEMPA,W     ;W ni TEMPA dan ayirish 
    BTFSS Z         ;agarda natija nollik bo‘lsa buyruqni 
                 ;o‘tkazib yuborish 
    GOTO    DD     ;DD belgiga o‘tish 
    RETURN 
  ; 
    INCLUDE   DELAY_E.ASM 
   ; 
Dasturning asosiy bajaradigan ishi A port holatini oldingi o‘qishdan biroz vaqt o‘tgandan so‘ng qayta o‘qish va uning oldingi qiymat bilan solishtirishdan iborat. W ga uzatiladigan H’0A’ = .10 konstantasi DELAY_E dasturostini chaqirishdan oldin 20 ms li vaqt qiymatini ushlanishini ta’minlaydi – bu odatda mexanik ulash vositalaridagi o‘tish jarayonlarini tugallanishiga yetaril vaqt hisoblanadi. Portning ishlatilmaydigan razryadlarini maskalash esa dasturostining ishlash ishonchliligini oshiradi. Ushlanishni hosil qiluvchi dasturostini chaqirishdan oldin qo‘riqchi taymerni nol holatga o‘tkazishning sababi, A portini ikkita so‘rov amali o‘rtasida MK ni nol holatga o‘tib qolishini oldini olish uchun kerak bo‘ladi.  
Endi dasturning ishini oldin 	yaratilgan dasturostilarning ba’zilaridan foydalangan holda ko‘rib chiqamiz. Dastur ishining maqsadi SB1 tugmani bosilish sonini hisoblash va uning natijasini yetti segmentli indikatorga o‘n oltilik kodda chiqarish bo‘lsin.  
  ;asosiy dastur 
    CLRF    COUNT3   ;bosishlar sanoq qurilmasini nolga o‘tqazish 
  LOOP 
    CLRWDT         ;qo‘riqchi taymerni nolga o‘tkazish  
    CALL    GET_RAD   ; GET_RAD dasturostini chaqirish   
    BTFSC TEMA,4      ;SB1 ni bosilishini tekshirish 
    GOTO    LOOP    ;agarda bosilmagan bo‘lsa LOOP belgiga  
                 ;qaytish 
    INCF    COUNT3,F  ;sanoq qurilmani inkrementi 
    MOVF    COUNT3,W  ;sanoq qurilma qiymatini ishchi registrga  
                 ;jo‘natish 
    CALL    SEV_SEG   ;dasturostini chaqirish   
  SEVEN_SEG 
     MOVWF   PORTB    ;PORTB ga Wni jo‘natish 
  TEST 
     CALL    GET_RAD   ;GET_RAD dasturostini chaqirish     
     BTFSS TEMPA,4      ;SB1 ni bosilishini tekshirish 
     GOTO   TEST     ;agarda hali ham bosilgan bo‘lsa TEST  
                 ;belgiga qaytish 
     GOTO   LOOP     ;LOOP belgiga qaytish      
  ; 
     INCLUDE    GET_RAD.ASM 
     INCLUDE    SEV_SEG.ASM 
  ; 
Ma’ruzada keltirilgan dasturlar hatto oddiy 19.1-chizmada keltirilgan maketning imkoniyatlarini kam qismini ham qamrab ololmaydi. Biroq ularni o‘zlashtirish PIC – kontrollerlaridan foydalanuvchilar uchun foydadan holi bo‘lmaydi albatta.  
Nazorat uchun savollar 
1. Laboratorya maketining sxemasini tushuntiring. 
2. Dastlabki dastur listingini tushintiring. 
3. Oddiy dastur tuzing va tushuntiring. 
4. Misol tariqasida vaqt ushlanishini hosil qiluvchi DELAY_C dasturostini tushuntiring.  

